
Obtaining these Instructions 
1. Either borrow a USB flash drive with this document and copy it to your machine or … 

2. Connect to our network  

a. A number of RJ-45 connectors (Ethernet cables) are available.  

b. Two wireless base stations are available. To help us balance the load:  

i. If you are on the left side of the room, please connect to GemStone1. 

ii. If you are on the right side of the room, please connect to GemStone2. 

c. Our network will assign a DHCP address to your machine. Please modify your 

network settings to avoid self-assigned IP addresses. 

d. Open a web browser on http://192.168.1.2/ and view this document. 

GemStone/S 64 Bit Setup Options 
3. Native install on 64-bit Linux or MacOSX 10.5. This is somewhat more complex, but does not 

require VMware. Follow these steps: 

a. Obtain a DVD or USB flash drive with the components. 

b. Make a directory /opt/gemstone. 

c. Copy GemTools.zip, GemStone64*.zip, and installGemstone* to that directory. 

d. Run the installGemstone* script appropriate to your OS. 

e. Once the server is started, use 'startSeaside_Hyper 8000' to start a web server.  

f. In a web browser go to http://localhost:8000/seaside. 

4. VMware Virtual Appliance. This is the approach most likely to get everything working 

quickly, but requires 64-bit hardware with virtualization support and an installed copy of 

VMware Server (free for Linux and Windows) or VMware Fusion (for Macintosh) along with a 

license key. The Virtual Appliance has a full install of 64-bit Linux, Apache, GemStone/S 64 

Bit, and other components (such as FastCGI). Follow these steps: 

a. Obtain a DVD or USB flash drive with the components 

b. Copy one of the VMware-guest64check* to a temp directory (if you are using 

Microsoft Windows, select the one ending with .exe; if you are using Linux, select 

the other one). 

c. Run the guest64check executable and verify that your hardware supports 

virtualization. If it does not, select another option. 

d. Reboot your machine into the BIOS setup utility. Ensure that hardware virtualization 

has been enabled (it is generally disabled by default). 

e. Copy GemTools.zip and GLASS-Appliance-1.0beta9.zip to a temp directory. Unzip the 

Appliance into your Virtual Machines directory and open it from VMware.  

f. Once started, you should see a web page with the guest's IP address. You can use 

the tools inside the guest or you can use the tools from your host system. 

5. Client-only connection. This approach should work for anyone that can connect to a 

network and run Squeak. The disadvantage is that everyone using this approach will be using 

the same server on a single laptop. 

a. Obtain a user ID, password, and port number from the instructor.  

b. Copy GemTools.zip from a DVD, from a USB flash drive, or from this link. 

http://192.168.1.2/


Introduction to Seaside and the Tools 
The goal of this exercise is to help you become familiar with the tools and with Seaside. 

a. Ensure that the server is running and then open a web browser on the server. For 

example, http://localhost:8000/seaside 

6. Explore the 'Counter Application', the most famous demo of Seaside. 

a. Select examples / counter, and try clicking the ++ link and the -- link.  

b. Right-click on the ++ link and select the menu item to open the link in a new tab or 

window.  

c. Click the ++ link a couple times in the second tab then return to the first tab. 

d. Guess what will be displayed when you click on the ++ link, and then try it. 

7. Explore the options made available with the Halos. 

a. Return to the second tab and click the 'Toggle Halos' link at the bottom. 

b. Click the 'S' link in the top right to see the HTML source.  

c. Click the 'R' link to return to the rendered component. 

8. Edit code using a web browser. 

a. Click on the notepad icon (the leftmost icon) to open a class browser. 

b. View the methods in the 'actions' category (click on 'actions' in the third column). 

c. View the code for the 'increase' method (click on 'increase' in the fourth column). 

d. Change the code to add two (2) instead of adding one (1). 

e. Click the [Accept] button at the bottom. 

f. Return to the first tab and click the ++ link to see the value increase by 2. 

 

9. Edit code in a debugger. 

a. In the second tab edit the code to add 'self halt.' (without the quotes) before the 

count assignment and save the method. 

b. In the first tab click the ++ link to bring up walkback. 

10. Note that by default Seaside shows the top five (5) frames on the stack.  

a. Click the <Full Stack> link and take a look at the stack. 

b. You could click the <Proceed> link to resume from the halt.  

c. Click the <Remote Debug> link to save this continuation in the database for future 

debugging. 

11. Open a debugger on this error: 

a. Launch GemTools 

b. Login to the database. 

i. In the Gem groupbox, edit the first text entry field (with the label '!tcp@') to 

have the hostname or IP address of the server.  

1. If you are running GemTools from within the VMware Virtual 

Appliance, then you can leave it as 'glass'. 

2. If you are running GemTools outside the VMWare Virtual Appliance, 

then you can get the IP address from the browser inside the 

appliance. 

3. If you are running the server with a native install, then you can use 

'localhost'. 

4. If you are using the shared server, then use '192.168.1.2'. 

http://localhost:8000/seaside


ii. If you have a local install (either native or VMware), the user ID should be 

left as 'DataCurator' with a password of 'swordfish'.  

iii. If you are using the shared server, then use the provided User ID and 

password. 

iv. Click the [Login] button. 

c. From the Transcript window, open a debugger: 

i. Click the [Debug] button. 

ii. Select [Okay] when asked to abort the transaction. This gives GemTools 

access to the latest database view. 

iii. If you are presented with a pop-up menu of errors, select the bottom one. 

iv. Scroll down till you get to 'WACounter | increase' with the red flag and 

select it (this will be about three pages if you haven't changed the window). 

12. Edit the code and proceed. 

a. Remove the 'self halt.' line and change the two (2) back to a one (1), then save your 

changes (using <Ctrl>+<S>). 

b. If the text area refreshes with the old text, click on a different line in the upper list 

and click back on the original line. This should show the new code. 

c. Click the [proceed] button in the debugger. 

d. Return to the first tab of your web browser and click the <resume> link. 

e. Note that the values now increase by one. 

 

13. Try various other pieces of the system (using your history to go back to the Dispatcher 

Viewer or starting over). 

14. Note that you get an error in the 'Mini Calendar' (tests / alltests).  

a. Click the <Remote Debug> link.  

b. Return to GemTools (in Squeak) and open a debugger. 

c. Select the last continuation with the label '^_MessageNotUnderstood 2010: No 

method was found for the selector <#'year'> when sent to <6>…" 

d. Select any stack frame in the list and click [inspect context] 

e. Examine various stack frames to get an idea of what Seaside does. Which frame has 

the error? 

15. The problem appears to be that the object in the instance variable 'month' in 

WAMiniCalendar is an integer and does not understand the message #'year'.  What classes 

implement the method #'year'? 

a. From the Transcript, click [Find Method…] 

b. Enter 'year' (without the quotes) and click Accept 

c. Select the first item in the list, 'year'. 

16. While any of these classes could be what is needed, it is also possible that a subclass would 

do. What subclasses exist for Timespan? 

a. From the Transcript, click [Browse…] 

b. Enter 'Timespan' and click Accept 

c. Note that the class 'Month' is a subclass of Timespan, so should implement #'year'. 



17. Where is the instance variable 'month' in WAMiniCalendar  referenced? 

a. From the Transcript, click [Browse…] 

b. Enter 'calendar' and click Accept 

c. Select 'WAMiniCalendar' 

d. In the second list, right click on 'WAMiniCalendar' and select 'chase variables'. (If this 

does not pop up a menu try the <ESC> key.) 

e. In the new window, click on 'month' in the first list. 

18. How does Date>>#month differ in Squeak from GemStone? 

a. In a Squeak workspace, inspect (Date today month) using <Ctrl>+<I> 

b. Do the same in a GemStone workspace (opened from the Transcript) 
It appears that in Squeak the method returns an instance of the class Month, while in 

GemStone the method returns an instance of the class SmallInteger. 

19. In the "Chasing Browser" (from #16 above), examine each of the five methods in 

WAMiniCalendar. Note that when you click on a method in the second list, a third list is 

added (look for the horizontal scroll bar). 

20. Which methods are reading the 'method' instance variable and which method(s) are setting 

the 'method' instance variable? 

21. What methods are available to construct a Month object? 

a. Select the System Browser on Month (it should still exist) 

b. Click the [Class] button in the second column 

c. Select the #'month:year:' method in the fourth column 

d. Examine the source code for this method 

22. Edit WAMiniCalendar>>#'initialize' so that the month assignment reads as follows (save with 

< Ctrl>+<S> or <right click>+<accept> or <Esc>+<accept>): 

month := WAValueHolder with: (Month  

month: Date today month  

year: Date today year). 

23. Test the changes. 

a. Return to your web browser, and start over with tests / alltests / Mini Calendar 

b. Return to Squeak and click the [Logout] button on the Transcript and note that the 

other windows close automatically.  

24. Clear the Object Log 

a. From the first Dispatcher Viewer, select tools / objectLog 

b. Review some of the information stored in the Object Log 

c. Delete individual lines, delete down to particular lines (the heading might say "up 

to" but it means from the top down to the selected one), and then groups (e.g., fatal 

or error) till they are all gone. 

This exercise has introduced you to running Seaside, editing code, and debugging an error. The 

Calendar error is a real-world example of one found when porting a Seaside application from Squeak 

to GemStone. The problem is that in these two dialects of Smalltalk the base class library has a 

different implementation of Date. GemStone keeps the month as an integer, while Squeak has a 

more sophisticated Month class. While there are not too many such inconsistencies, there are a few 

and knowing what to watch for can make it easier to port a Seaside application from Squeak to 

GemStone. 



A First Seaside Component 
1. Log in to your database with GemTools (in Squeak).  

2. Open a Class Browser: 

a. From the Transcript, open a Class Browser by clicking [Browse…] 

b. Enter 'WAComponent' and click [Accept] 

3. Create yet another counter class. In the text area of the Class Browser, replace the current 

class definition with the following and then save the text (<Ctrl>+<S>): 

WAComponent subclass: 'WebCounter' 

instVarNames: #( count) 

classVars: #() 

classInstVars: #() 

poolDictionaries: #[] 

inDictionary: '' 

category: 'GLASS' 

4. Add an #'initialize' method to the instance side of WebCounter. Click on the method 

category '---all---' and then replace the text entry field with the following: 

initialize 

super initialize. 

count := 0. 

5. Create an #'increase' method: 

increase 

count := count + 1. 

6. Create a #'decrease' method: 

decrease 

count := count - 1. 

7. Now we define the method #”renderContentOn:' to display the counter as heading. Seaside 

will call such a method when needed and display it in the web browser. In the following 

method we just say that we want to display the value of the variable count using a heading 

HTML tag.  

renderContentOn: html  

html heading: count. 

 

8. Now we should register a component as an application so that we can access it directly from 

the url path that will be associated with it. To register a component as an application, we 

send the message registerAsApplication: to the class we created and specify a path string 

that will be used to access the component from the web browser. The following code 

snippet registers the component WebCounter as the application named 'WebCounter' and 

can be executed in the Transcript.  

WebCounter registerAsApplication: 'WebCounter'. 



 

9. Note that this expression can also be added in the class #'initialize' method which is invoked 

when the class is loaded in memory. In the Class Browser, switch to the class-side methods 

(click the [Class] button), select the method category '---all---', and then enter the following 

method: 

initialize 

super initialize. 

self registerAsApplication: 'WebCounter'. 

10. Commit the changes. 

a. From the Transcript, click [Commit] 

11. Now you can launch the application in your web browser by going to: 

http://localhost:8000/seaside/WebCounter  

12. Now we can add some actions by defining callbacks attached to anchors. A callback is a piece 

of code that will be executed when a link is clicked. Modify the instance method 

#'renderContentOn:' as follows: 

renderContentOn: html 

 html heading: count. 

 html anchor 

  callback: [self increase]; 

  with: '++'. 

 html space. 

 html anchor 

  callback: [self decrease]; 

  with: '--'. 

13. Refresh the page in your browser and click the ++ link a few times. Then try opening the ++ 

link in a new tab (or window). Click the ++ link a few times in the second tab. Return to the 

first tab and guess what will happen when you click the ++ link. 

14. This counter application does not behave in the way the famous Seaside demo counter 

works because we have not identified the counter as something that should have its state 

saved. Add a #'states' method on the instance side of WebCounter: 

states 

^Array with: self. 

15. Restart the application in your web browser by going to: 

http://localhost:8000/seaside/WebCounter  

16. Try the exercise again, opening the ++ link in a new tab and see if the two web pages now 

share state. 

  

http://localhost:8000/seaside/WebCounter
http://localhost:8000/seaside/WebCounter


Los Boquitas 
Your child has been recruited to play on the best youth football (soccer for you Americans) team in 

the region, Los Boquitas. All the parents are expected to be involved and instead of coaching on the 

field, you have agreed to create a web site to manage some team information. After a review of 

various technologies, you have decided to try GLASS—GemStone, Linux, Apache, Seaside, and 

Smalltalk. 

Following is a picture of the basic structure, showing various pieces of a web site (header, sidebar, 

main, image, footer), that we will build. 

 

 

17. First we will define a home page for the site. We will use 'LB' (for Los Boquitas) as the prefix 

for our classes.  

WAComponent subclass: 'LBHome' 

 instVarNames: #() 

 classVars: #() 

 classInstVars: #() 

 poolDictionaries: #[] 

 category: 'LosBoquitas' 

a. To identify the component as being a root component , add this method to the class 

side of LBHome: 

canBeRoot 

 

 ^true. 



b. To provide a description that will show on the dispatcher page, add this class-side 

method to LBHome (select 'Override' when prompted to confirm the method name): 

description 

 

 ^'Keep track of children''s football (soccer) team'. 

c. To provide a simple way to register the application, add this class-side method: 

initialize 

" 

 LBHome initialize. 

" 

 (self registerAsApplication: 'boquitas') 

  preferenceAt: #sessionClass  

  put: WAExpirySession. 

d. Now we are ready to add some content. Add this instance-side method to LBHome: 

renderContentOn: canvas 

 

 canvas heading  

  level: 1; 

  with: 'Los Boquitas Soccer Team'. 

e. Now initialize your component by executing the following in a workspace (like the 

Transcript window): 

LBHome initialize. 

f. Finally, start at the dispatcher page, http://localhost:8000/seaside, and follow the 

link to boquitas. If your application is not visible, then go back and see if you put the 

proper methods on the class side of LBHome. If they are on the instance side, then 

things will not work. 

18. Adding an image. 

a. Modify the render method in LBHome as follows: 

renderContentOn: canvas 

 

 canvas heading  

  level: 1; 

  with: 'Los Boquitas Soccer Team'. 

 canvas image 

  url: '/images/youthSoccer1.jpg'; 

  yourself. 

http://localhost:8000/seaside


b. Try viewing the page and notice that the image does not display. Depending on your 

browser, a placeholder might be displayed. At a minimum we need some alternate 

text to be displayed when the image is missing. Modify the render method again: 

renderContentOn: canvas 

 

 canvas heading  

  level: 1; 

  with: 'Los Boquitas Soccer Team'. 

 canvas image 

  altText: 'children playing soccer'; 

  url: '/images/youthSoccer1.jpg'; 

  yourself.  

c. Try viewing the page and verify that the alternate text is displayed. Now we will 

update the link to a site that does have the picture. Modify the render method: 

renderContentOn: canvas 

 

 canvas heading  

  level: 1; 

  with: 'Los Boquitas Soccer Team'. 

 canvas image 

  altText: 'children playing soccer'; 

  url: 'http://192.168.1.2/images/youthSoccer1.jpg'; 

  yourself. 

d. View the page and verify that the image shows. 

19. Adding a page title. 

a. Note that the page is simply titled "Seaside" rather than something more 

descriptive.  Add the following instance-side method to LBHome: 

updateRoot: anHtmlRoot 

 

 super updateRoot: anHtmlRoot. 

 anHtmlRoot title: 'Los Boquitas'. 

b. View the page again and note that it now has a title. 



20. Creating multiple areas. 

a. The typical web site has a header, a side-bar, a main content area, and a footer. We 

will now add these pieces by modifying the render method again: 

renderContentOn: canvas 

 

 canvas div 

  id: 'allcontent'; 

  with: [ 

   canvas div 

    id: 'header'; 

    class: 'section'; 

    with: [ 

     canvas heading 

      level1; 

      with: 'Los Boquitas Soccer Team'; 

      yourself. 

    ]; 

    yourself. 

   canvas div 

    id: 'main'; 

    class: 'section'; 

    with: [ 

     canvas image 

      altText: 'children playing soccer'; 

      url: 'http://192.168.1.2/images/youthSoccer1.jpg'; 

      yourself. 

    ]; 

    yourself. 

   canvas div 

    id: 'sidebar'; 

    class: 'section'; 

    with: [ 

     canvas heading 

      level2; 

      with: 'Sidebar'; 

      yourself. 

    ]; 

    yourself. 

   canvas div 

    id: 'footer'; 

    class: 'section'; 

    with: [ 

     canvas text: 'Copyright (c) ' , Date today year printString. 

    ]; 

    yourself. 

  ]. 

b. View the page and confirm that the various pieces exist. Note, however, that they do 

not have any formatting. Until recently, the typical way of doing page layout was to 

use a table. This approach is no longer recommended and the better approach is to 

use Cascading Style Sheets (CSS). 



21. Adding CSS to the site. Modify the following method to add a style sheet reference: 

updateRoot: anHtmlRoot 

 

 super updateRoot: anHtmlRoot. 

 anHtmlRoot title: 'Los Boquitas'. 

 anHtmlRoot link 

  type: 'text/css'; 

  beStylesheet; 

  addAll; 

  url: 'http://192.168.1.2/css/boquitas.css'; 

  yourself. 

22. Let's refactor the render code so that it is more modular. Add the following methods: 

renderHeaderOn: canvas 

 

 canvas div 

  id: 'header'; 

  class: 'section'; 

  with: [ 

   canvas heading 

    level1; 

    with: 'Los Boquitas Soccer Team'; 

    yourself. 

  ]; 

  yourself. 

 

renderMainOn: canvas 

 

 canvas div 

  id: 'main'; 

  class: 'section'; 

  with: [ 

   canvas image 

    altText: 'children playing soccer'; 

    url: 'http://192.168.1.2/images/youthSoccer1.jpg'; 

    yourself. 

  ]; 

  yourself. 

 



renderSidebarOn: canvas 

 

 canvas div 

  id: 'sidebar'; 

  class: 'section'; 

  with: [ 

   canvas heading 

    level2; 

    with: 'Sidebar'; 

    yourself. 

  ]; 

  yourself. 

 

renderFooterOn: canvas 

 

 canvas div 

  id: 'footer'; 

  class: 'section'; 

  with: [ 

   canvas text: 'Copyright (c) ' , Date today year printString. 

  ]; 

  yourself. 

a. And now modify the renderContentOn: method to call these new methods: 

renderContentOn: canvas 

 

 canvas div 

  id: 'allcontent'; 

  with: [ 

   self 

    renderHeaderOn: canvas; 

    renderMainOn: canvas; 

    renderSidebarOn: canvas; 

    renderFooterOn: canvas; 

    yourself. 

  ]. 

23. The next task is to add the ability to display a schedule of events. But first, we need to have 

some events to display.  

a. We will start by defining an event class: 

Object subclass: 'LBEvent' 

 instVarNames: #( date location description) 

 classVars: #() 

 classInstVars: #( events) 

 poolDictionaries: #[] 

 inDictionary: '' 

 category: 'LosBoquitas' 



b. Next we will create accessors for the instance variables by executing the following in 

a workspace (such as the Transcript): 

LBEvent compileMissingAccessingMethods. 

c. Add a class-side method to access the events: 

events 

 

 events isNil ifTrue: [events := SortedCollection new]. 

 ^events. 

d. Add an instance-side method to support sorting the events: 

<= anEvent 

 

 ^self date <= anEvent date. 

e. Add an initialize method to ensure that something is in each instance variable: 

initialize 

 

 super initialize. 

 date := Date today. 

 location := 'field'. 

 description := 'practice'. 

f.  Add a class-side method to create some sample events: 

createEvents 

 

 self events 

  add: (self new 

   date: (Date today addDays: 0); 

   location: 'Clubhouse'; 

   description: 'Registration'; 

   yourself); 

  add: (self new 

   date: (Date today addDays: 1); 

   yourself); 

  add: (self new 

   date: (Date today addDays: 2); 

   location: 'Memorial Park'; 

   description: 'Game with Metro'; 

   yourself); 

  yourself. 

g. Call the createEvents method: 

LBEvent createEvents. 



24. Now we will define a component to display the schedule: 

WAComponent subclass: 'LBScheduleComponent' 

 instVarNames: #(events) 

 classVars: #() 

 classInstVars: #() 

 poolDictionaries: #[] 

 inDictionary: '' 

 category: 'LosBoquitas' 

a. Initially we will treat this as a stand-alone component (or application). Later, we will 

incorporate it into the main application. For now, add the following class-side 

methods: 

canBeRoot 

 

 ^true. 

 

description 

 

 ^'Schedule for Los Boquitas'. 

 

initialize 

" 

 LBScheduleComponent initialize. 

" 

 (self registerAsApplication: 'boquitas-schedule') 

  preferenceAt: #sessionClass  

  put: WAExpirySession. 

b. Initialize the component by executing the initialize method. You can do this by 

clicking in the line to be executed and pressing <Ctrl>+<D>, or by pressing <Esc> and 

selecting 'Do It' from the menu, or by right-clicking and selecting 'Do It' from the 

menu.  

c. Afterwards, commit the initialization to the database by clicking the [Commit] 

button on the Transcript. 

d. Add a place-holder render method: 

renderContentOn: canvas 

 

 canvas text: 'This is the LBScheduleComponent'. 

e. In a web browser, navigate to the dispatcher and confirm that the new component is 

in the list and that it displays the text provided. 



25. Now we will add a real display capability to the component.  

a. Add an initialize method to create a table report: 

initialize 

 

 | columns | 

 super initialize. 

 columns := Array 

  with: (WAReportColumn new 

   title: 'Date'; 

   selector: #date; 

   clickBlock: nil; 

   yourself) 

  with: (WAReportColumn new 

   title: 'Location'; 

   selector: #location; 

   clickBlock: nil; 

   yourself) 

  with: (WAReportColumn new 

   title: 'Description'; 

   selector: #description; 

   clickBlock: nil; 

   yourself). 

 events := WATableReport new 

  columns: columns; 

  rowPeriod: 1; 

  yourself. 

b. Now modify the render method to show the table: 

renderContentOn: canvas 

 

 events rows: LBEvent events. 

 canvas render: events. 

c. Starting from the dispatcher in a web browser, view the schedule component and 

confirm that it shows three rows of three columns. 

26. Now we will update our main application to make room for a child component. 

a. Define a new schema for LBHome to add an instance variable to hold the 

component being displayed in the main region: 

WAComponent subclass: 'LBHome' 

 instVarNames: #(mainArea) 

 classVars: #() 

 classInstVars: #() 

 poolDictionaries: #[] 

 inDictionary: '' 

 category: 'LosBoquitas' 



b. Modify LBMain>>#renderSidebarOn: to change the heading: 

renderSidebarOn: canvas 

 

 canvas div 

  id: 'sidebar'; 

  class: 'section'; 

  with: [ 

   canvas heading 

    level2; 

    with: 'Menu'; 

    yourself. 

  ]; 

  yourself. 

c. Return to your web browser and display the home page. Note that the sidebar text 

has not changed. This is because GemStone/S allows multiple versions of a class and 

the old version of the class is still the registered component. When you create a new 

version of a registered application, you need to re-register it to have the new version 

take effect. Execute (and commit) the following to register the new version of the 

class: 

LBHome initialize. 

d. Return to your browser and display the home page. It should have the new text now 

('Menu' instead of 'Sidebar'). 

27. Add a menu to the sidebar. 

a. Modify the sidebar render method as follows: 

renderSidebarOn: canvas 

 

 canvas div 

  id: 'sidebar'; 

  class: 'section'; 

  with: [ 

   canvas heading 

    level2; 

    with: 'Menu'; 

    yourself. 

   canvas anchor 

    callback: [mainArea := LBScheduleComponent new]; 

    with: 'Events'; 

    yourself. 

  ]; 

  yourself. 

b. View the home page in a browser and confirm that the <Events> link is present. 

Clicking on it does not have any impact, but it is there! 



c. We want the render method to use the mainArea component if it exists; otherwise, 

the image will be displayed. Modify the renderMainOn: method as follows: 

renderMainOn: canvas 

 

 canvas div 

  id: 'main'; 

  class: 'section'; 

  with: [ 

   mainArea notNil ifTrue: [ 

    canvas render: mainArea. 

   ] ifFalse: [ 

    canvas image 

     altText: 'children playing soccer'; 

     url: 'http://192.168.1.2/images/youthSoccer1.jpg'; 

     yourself. 

   ]. 

  ]; 

  yourself. 

d. View the home page in a web browser and confirm that the event list displays. 

e. We now want a way to return to the home page. Modify the sidebar render method 

as follows: 

renderSidebarOn: canvas 

 

 canvas div 

  id: 'sidebar'; 

  class: 'section'; 

  with: [ 

   canvas heading 

    level2; 

    with: 'Menu'; 

    yourself. 

   canvas anchor 

    callback: [mainArea := nil]; 

    with: 'Home'; 

    yourself. 

   canvas break. 

   canvas anchor 

    callback: [mainArea := LBScheduleComponent new]; 

    with: 'Events'; 

    yourself. 

  ]; 

  yourself. 

f. View the application in a web browser and confirm that you can switch between the 

image and the schedule. 



28. We would like to be able to edit events. We will start with deleting an event. 

a. Edit LBScheduleComponent>>initialize to add a delete option: 

initialize 

 

 | columns | 

 super initialize. 

 columns := Array 

  with: (WAReportColumn new 

   title: 'Date'; 

   selector: #date; 

   clickBlock: nil; 

   yourself) 

  with: (WAReportColumn new 

   title: 'Location'; 

   selector: #location; 

   clickBlock: nil; 

   yourself) 

  with: (WAReportColumn new 

   title: 'Description'; 

   selector: #description; 

   clickBlock: nil; 

   yourself) 

  with: (WAReportColumn new 

   title: 'Action'; 

   valueBlock: [:anEvent | 'delete']; 

   clickBlock: [:anEvent | self delete: anEvent]; 

   yourself). 

 events := WATableReport new 

  columns: columns; 

  rowPeriod: 1; 

  yourself. 

b. If you return to your web browser and refresh, the new column will likely not 

appear. This is because the component is still holding an instance of WATableReport 

that was initialize with only three columns. To see the new table you need to click on 

<Events>. 

c. If you click on a <delete> link now, you should get an error "Components not found 

while processing callbacks: anArray." As suggested in the possible causes, this is 

because we have not implemented children correctly. Add the following method to 

LBScheduleComponent: 

children 

 

 ^Array with: events. 



d. Try refreshing your web browser and note that the same error occurs. This is 

because we actually have two levels of components. Add a similar method to 

LBHome: 

children 

 

 ^Array with: mainArea. 

e. Try refreshing your web browser and note that the error has now changed to a 

MessageNotUnderstood because we have not implemented the #delete: method in 

LBScheduleComponent: 

delete: anEvent 

 

 LBEvent events remove: anEvent. 

f. Try refreshing your web browser and note that one of the events has been removed. 

29. Editing an existing event. 

a. Define a new component to edit events: 

WAComponent subclass: 'LBEventEditor' 

 instVarNames: #( event) 

 classVars: #() 

 classInstVars: #() 

 poolDictionaries: #[] 

 inDictionary: '' 

 category: 'LosBoquitas' 

b. Add a render method to show that we are displaying the component: 

renderContentOn: canvas 

 

 canvas text: 'LBEventEditor'. 

c. Create accessors for the instance variable(s): 

LBEventEditor compileMissingAccessingMethods. 

 



d. Now we will modify LBScheduleComponent>>initialize to call the new component 

from a clickBlock on the date column: 

initialize 

 

 | columns | 

 super initialize. 

 columns := Array 

  with: (WAReportColumn new 

   title: 'Date'; 

   selector: #date; 

   clickBlock: [:anEvent | self edit: anEvent]; 

   yourself) 

  with: (WAReportColumn new 

   title: 'Location'; 

   selector: #location; 

   clickBlock: nil; 

   yourself) 

  with: (WAReportColumn new 

   title: 'Description'; 

   selector: #description; 

   clickBlock: nil; 

   yourself) 

  with: (WAReportColumn new 

   title: 'Action'; 

   valueBlock: [:anEvent | 'delete']; 

   clickBlock: [:anEvent | self delete: anEvent]; 

   yourself). 

 events := WATableReport new 

  columns: columns; 

  rowPeriod: 1; 

  yourself. 

e. In your web browser, click on the <Events> link to show the date field as a link. Click 

on any date to get an error (because the #edit: method is not yet implemented). Add 

it as follows: 

edit: anEvent 

 

 | editor answer | 

 editor := LBEventEditor new 

  event: anEvent; 

  yourself. 

 answer := self call: editor. 

 answer 

  ifTrue: [self inform: 'Edits were saved'] 

  ifFalse: [self inform: 'Edits were cancelled']. 

f. You should be able to refresh your browser and see that the schedule list is replaced 

with the event editor component (which simply displays some text). 



30. Add true editing to the editor. 

a. Modify LBEventEditor>>#renderContentOn: as follows: 

renderContentOn: canvas 

 

 canvas form: [ 

  canvas table: [ 

   canvas tableBody: [ 

    canvas tableRow: [ 

     canvas tableHeading: 'Date:'. 

     canvas tableData: [ 

      canvas dateInput 

       value: event date; 

       callback: [:value | event date: value]. 

     ]. 

    ]. 

    canvas tableRow: [ 

     canvas tableHeading: 'Location:'. 

     canvas tableData: [ 

      canvas textInput 

       value: event location; 

       callback: [:value | event location: value]. 

     ]. 

    ]. 

    canvas tableRow: [ 

     canvas tableHeading: 'Description:'. 

     canvas tableData: [ 

      canvas textArea 

       value: event description; 

       callback: [:value | event description: value]. 

     ]. 

    ]. 

    canvas tableRow: [ 

     canvas tableData: [ 

      canvas  cancelButton 

       callback: [self answer: false]; 

       with: 'Cancel'. 

     ]. 

     canvas tableData. 

     canvas tableData: [ 

      canvas  submitButton 

       callback: [self answer: true]; 

       with: 'Save'. 

     ]. 

    ]. 

   ]. 

  ]. 

 ]. 

b. Try this component in your web browser. It should be possible to edit the fields and 

save or cancel the edits. Cancelled edits should not be persisted. 



c. Note how a table is used to lay out the form. This is a fairly typical approach because 

it allows labels and data entry fields to be positioned relatively nicely. Note how we 

moved the [Save] button to the right by inserting an extra table cell in the last row. 

d. Edit LBEventEditor>>#renderContentOn: to remove the table-based formatting: 

renderContentOn: canvas 

 

 canvas form 

  id: 'eventEditor'; 

  with: [ 

   canvas label 

    id: 'dateLabel'; 

    with: 'Date:'. 

   canvas dateInput 

    id: 'date'; 

    value: event date; 

    callback: [:value | event date: value]. 

   canvas label 

    id: 'locationLabel'; 

    with: 'Location:'. 

   canvas textInput 

    id: 'location'; 

    value: event location; 

    callback: [:value | event location: value]. 

   canvas label 

    id: 'descriptionLabel'; 

    with: 'Description:'. 

   canvas textArea 

    id: 'description'; 

    value: event description; 

    callback: [:value | event description: value]. 

   canvas cancelButton 

    id: 'cancel'; 

    callback: [self answer: false]; 

    with: 'Cancel'. 

   canvas submitButton 

    id: 'submit'; 

    callback: [self answer: true]; 

    with: 'Save'. 

  ]. 

e. View this in a browser and observe that the layout is simply one field after another. 



f.  The "proper" way to style a web page is with CSS. While this is generally done in a 

separate file served by your web server, it can be incorporated into your Seaside 

component. Add the following method to LBEventEditor: 

style 

 

^'#eventEditor #date { margin-left: 4em; }  

#eventEditor #date-year { margin-right: 10em; } 

#eventEditor .dateGroup { margin: 0em; padding: 0em; } 

#eventEditor .dateGroup input { margin-left: 1.1em; } 

#eventEditor #location { margin-left: 2em; margin-right: 15em; } 

#eventEditor #description { margin-left: 0.8em; width: 25em; } 

#eventEditor #cancel { margin-left: 7em; } 

.dateGroup { border: 0; position: relative; } 

.dateGroup a.calendarAnchor img { border: 0; } 

.calendarContainer { display: none; font-size: small;  

 margin-left: 5em; z-index: 1; }'. 

g. Refresh the page in your web browser, and note that the positioning is now 

controlled by the CSS. We have separated the text markup (HTML) from the style 

(CSS). This is considered a much better way to build web sites. 

h. Note that this is simply an example. I'm not (yet) much of a CSS expert, and the 

month list does not drop down as it should. You can use the arrow keys to change 

months, but that isn't really proper. If anyone knows how to fix this with CSS, please 

let me know. 

31. Editing CSS from the web. 

a. While viewing the LBEventEditor component in a web browser, click the <Toggle 

Halos> link at the bottom of the page.  

b. Then click on the CSS Style Editor icon (the three colored circles with a paint brush) 

in the LBEventEditor title.  

c. This will bring up a text editor with the style string. You can edit and save the string. 

Try changing a value, saving, and then view the change in your code browser.  

d. Note that you might need to abort or commit to get a fresh view of the database 

that includes the changed method. 

32. Adding a new event. 

a. Modify LBScheduleComponent>>#renderContentOn: to add an <Add> link: 

renderContentOn: canvas 

 

 events rows: LBEvent events. 

 canvas render: events. 

 canvas anchor 

  callback: [self add]; 

  with: 'Add'; 

  yourself. 

b. Try it out and note that you get a walkback because the add method is not 

implemented. 



c. Add the following method: 

add 

 

 | event editor | 

 event := LBEvent new. 

 editor := LBEventEditor new 

  event: event; 

  yourself. 

 (self call: editor) ifTrue: [ 

  LBEvent events add: event. 

 ]. 

d. Refresh your browser and try adding an event. Try opening the editor but cancelling 

the new event.  

e. Note how we are reusing a component—to add and to edit. The component doesn't 

know how it is being used which provides for good encapsulation. 

f. Note also that the answer is useful in this case. If the user pressed the Cancel 

button, we don't want to add the new event. 

33. Now let us look at supporting user login. This will allow us to separate users who can view 

some pages from users who can edit all pages. 

a. Create a class to model the user: 

Object subclass: 'LBUser' 

 instVarNames: #( id name password) 

 classVars: #() 

 classInstVars: #( users) 

 poolDictionaries: #[] 

 inDictionary: '' 

 category: 'LosBoquitas' 

b. Create accessors: 

LBUser compileMissingAccessingMethods. 

c. Add the following instance-side methods: 

initialize 

 

 super initialize. 

 id := ''. 

 name := ''. 

 password := ''. 

 

<= aUser 

 

 ^self id <= aUser id. 



d. Add the following class-side method: 

users 

 

 users isNil ifTrue: [ 

  users := SortedCollection with: (self new  

   id: 'admin'; 

   name: 'Site Administrator'; 

   password: 'passwd'; 

   yourself). 

 ]. 

 ^users. 

34. Create a Seaside session to hold session information, including the user. 

a. Define the following class and add accessor methods (do you remember how?): 

WAExpirySession subclass: 'LBSession' 

 instVarNames: #( user requestLogout) 

 classVars: #() 

 classInstVars: #() 

 poolDictionaries: #[] 

 inDictionary: '' 

 category: 'LosBoquitas' 

b. Modify LBHome class>>initialize to use this new session class: 

initialize 

" 

 LBHome initialize. 

" 

 (self registerAsApplication: 'boquitas') 

  preferenceAt: #sessionClass  

  put: LBSession. 

c. Initialize LBHome so that it uses the new class. 

35. Add a login component to the home page: 

a. Create a new login component: 

WAComponent subclass: 'LBLoginComponent' 

 instVarNames: #( userID password) 

 classVars: #() 

 classInstVars: #() 

 poolDictionaries: #[] 

 inDictionary: '' 

 category: 'LosBoquitas' 

b. Add a render method to show that it is being called. 

renderContentOn: canvas 

 

 canvas text: 'LBLoginComponent'. 



c. Create a new version of the LBHome class that includes an instance variable for the 

login component: 

WAComponent subclass: 'LBHome' 

 instVarNames: #( mainArea login) 

 classVars: #() 

 classInstVars: #() 

 poolDictionaries: #[] 

 inDictionary: '' 

 category: 'LosBoquitas' 

d. Initialize LBHome so that the new version of the class is registered. 

e. Add a method to LBHome to create the login component: 

initialize 

 

 super initialize. 

 login := LBLoginComponent new. 

f.  Edit the children method to include the new component: 

children 

 

 ^Array  

  with: mainArea 

  with: login. 

g. Edit the sidebar render method to render the new component: 

renderSidebarOn: canvas 

 

 canvas div 

  id: 'sidebar'; 

  class: 'section'; 

  with: [ 

   canvas heading 

    level2; 

    with: 'Menu'; 

    yourself. 

   canvas anchor 

    callback: [mainArea := nil]; 

    with: 'Home'; 

    yourself. 

   canvas break. 

   canvas anchor 

    callback: [mainArea := LBScheduleComponent new]; 

    with: 'Events'; 

    yourself. 

   canvas break. 

   canvas render: login. 

  ]; 

  yourself. 



h. View the application and verify that the LBLoginComponent is being displayed. If 

not, try reinitializing the application (LBHome initialize). 

36. Add the login form to the component. 

a. Edit LBLoginComponent>>#renderContentOn: as follows: 

renderContentOn: canvas 

 

 self session user isNil 

  ifTrue: [self renderLoginOn: canvas] 

  ifFalse: [self renderLogoutOn: canvas]. 

b. Add the following method: 

renderLoginOn: canvas 

 

 canvas form: [ 

  canvas break. 

  canvas text: 'User: '. 

  canvas break. 

  canvas textInput 

   value: ''; 

   callback: [:value | userID := value]; 

   yourself. 

  canvas break. 

  canvas text: 'Password: '. 

  canvas break. 

  canvas passwordInput 

   value: ''; 

   callback: [:value | password := value]; 

   yourself. 

  canvas break. 

  (userID notNil and: [userID notEmpty]) ifTrue: [ 

   canvas text: 'Login failed!'; break. 

  ]. 

  canvas submitButton 

   callback: [self login]; 

   with: 'Login'; 

   yourself. 

 ]. 

c. Note that we are using the break message to generate a <br> tag in the HTML. 

Technically, this is formatting that should be replaced with some CSS as we did 

earlier. 

d. Refresh the application in your web browser and try clicking on the [Login] button. 

You should get an error since #login is not yet implemented. 



37. Add code to process the login. 

a. Add the following method: 

login 

 

 | user | 

 user := LBUser users 

  detect: [:each | each id = userID and: [each password = password]] 

  ifNone: [nil]. 

 user notNil ifTrue: [ 

  self session user: user. 

  ^self. 

 ]. 

b. Now try the application again. If you give a wrong user ID/password, you should get 

a message displayed with that information. 

c. If you give the correct user ID and password, you should get a walkback due to 

missing the logout render code. 

38. Add code to render and process the logout. 

a. Add the following method: 

renderLogoutOn: canvas 

 

 canvas anchor 

  callback: [self logout]; 

  with: 'Logout ' , self session user name; 

  yourself. 

b. Refresh the application and the logout link should display. If you click on the link you 

should get a walkback since the logout method is not implemented. 

c. Add the following method: 

logout 

 

 self session user: nil. 

 userID := nil. 

 password := nil. 

d. Refresh the application and the logout should be complete with the login form 

showing again. 



39. Restrict some features to logged-in users. 

a. Modify LBScheduleComponent>>#renderContentOn: as follows: 

renderContentOn: canvas 

 

 events rows: LBEvent events. 

 canvas render: events. 

 self session user notNil ifTrue: [ 

  canvas anchor 

   callback: [self add]; 

   with: 'Add'; 

   yourself. 

 ]. 

b. Modify LBScheduleComponent>>#initialize as follows: 

initialize 

 

 | editBlock deleteBlock columns | 

 self session user notNil ifTrue: [ 

  editBlock := [:anEvent | self edit: anEvent]. 

  deleteBlock := [:anEvent | self delete: anEvent]. 

 ]. 

 super initialize. 

 columns := Array 

  with: (WAReportColumn new 

   title: 'Date'; 

   selector: #date; 

   clickBlock: editBlock; 

   yourself) 

  with: (WAReportColumn new 

   title: 'Location'; 

   selector: #location; 

   clickBlock: nil; 

   yourself) 

  with: (WAReportColumn new 

   title: 'Description'; 

   selector: #description; 

   clickBlock: nil; 

   yourself) 

  with: (WAReportColumn new 

   title: 'Action'; 

   valueBlock: [:anEvent | 'delete']; 

   clickBlock: deleteBlock; 

   yourself). 

 events := WATableReport new 

  columns: columns; 

  rowPeriod: 1; 

  yourself. 



 

c. View the application and note how the event list drawing is affected by the login 

state. Because some of the fields are set when the component is initialized, the 

rendering is not updated until you click the <Home> link and back to the <Events> 

link. 

d. Modify LBHome>>#renderSidebarOn: so that the login/logout feature is available 

only when the home link is selected: 

renderSidebarOn: canvas 

 

 canvas div 

  id: 'sidebar'; 

  class: 'section'; 

  with: [ 

   canvas heading 

    level2; 

    with: 'Menu'; 

    yourself. 

   canvas anchor 

    callback: [mainArea := nil]; 

    with: 'Home'; 

    yourself. 

   canvas break. 

   canvas anchor 

    callback: [mainArea := LBScheduleComponent new]; 

    with: 'Events'; 

    yourself. 

   canvas break. 

   mainArea isNil ifTrue: [ 

    canvas render: login. 

   ]. 

  ]; 

  yourself. 

e. Try the application and note how the login component is displayed or not displayed. 

  



40. Now for some Javascript! We are going to add a calendar widget to the event editor. This 

widget is based on the Yahoo User Interface (YUI) Library. See 

http://developer.yahoo.com/yui/calendar/ for more information and other examples. 

a. Create a new version of the event editor class that includes room for some 

Javascript code: 

WAComponent subclass: 'LBEventEditor' 

 instVarNames: #( event calendarScript) 

 classVars: #() 

 classInstVars: #() 

 poolDictionaries: #[] 

 inDictionary: '' 

 category: 'LosBoquitas' 

b. Add the following methods to LBEventEditor: 

updateRoot: anHtmlRoot 

 

 super updateRoot: anHtmlRoot. 

 self addYahooCalendarTo: anHtmlRoot. 

 

addYahooCalendarTo: anHtmlRoot 

  

 anHtmlRoot link 

  type: 'text/css'; 

  beStylesheet; 

  addAll; 

  "url: 

'http://yui.yahooapis.com/2.5.0/build/calendar/assets/skins/sam/calendar.cs

s';" 

  url: 'http://192.168.1.2/css/calendar.css'; 

  yourself. 

 anHtmlRoot script 

  beJavascript; 

  "url: 'http://yui.yahooapis.com/2.5.0/build/yahoo-dom-event/yahoo-

dom-event.js';" 

  url: 'http://192.168.1.2/scripts/yahoo-dom-event.js'; 

  yourself. 

 anHtmlRoot script 

  beJavascript; 

  "url: 'http://yui.yahooapis.com/2.5.0/build/calendar/calendar-

min.js';" 

  url: 'http://192.168.1.2/scripts/calendar-min.js'; 

  yourself. 

 

http://developer.yahoo.com/yui/calendar/


calendarScriptFor: idString title: titleString date: aDate 

 

 | pageDate selected | 

 (aDate isNil or: [(aDate isKindOf: String) and: [aDate isEmpty]]) 

ifTrue: [ 

  pageDate := (Date today asStringUsingFormat: #(1 2 3 $/ 1 1 )) 

copyFrom: 4 to: 10. 

  selected := Date today asStringUsingFormat: #(2 1 3 $/ 1 1 ). 

 ] ifFalse: [(aDate isKindOf: String) ifTrue: [ 

  pageDate := (aDate copyFrom: 1 to: 3) , (aDate copyFrom: 7 to: 10). 

  selected := aDate. 

 ] ifFalse: [ 

  pageDate := (aDate asStringUsingFormat: #(1 2 3 $/ 1 1 )) copyFrom: 4 

to: 10. 

  selected := aDate asStringUsingFormat: #(2 1 3 $/ 1 1 ). 

 ]]. 

^' YAHOO.example.calendar.' , idString ,  

  ' = new YAHOO.widget.Calendar("' , idString ,  

  '","cal' , idString ,  

  '", { navigator:true, title:"' , titleString , ' on:", pagedate:"' , 

pageDate , '", selected:"' , selected , '", close:true } ); 

 YAHOO.example.calendar.' , idString , '.render(); 

 YAHOO.util.Event.addListener("show' , idString ,  

  '", "click", YAHOO.example.calendar.' , idString ,  

  '.show, YAHOO.example.calendar.' , idString ,  

  ', true); 

 function handleSelect' , idString , '(type,args,obj) {  

  var dates = args[0];  

  var date = dates[0];  

  var year = date[0], month = "0" + date[1], day = "0" + date[2];  

  month = month.substring(month.length-2, month.length); 

  day = day.substring(day.length-2, day.length); 

  var txtDate = document.getElementById("' , idString , '");  

  txtDate.value = month + "/" + day + "/" + year;  

  YAHOO.example.calendar.' , idString , '.hide(); 

 }  

 YAHOO.example.calendar.' , idString , 

'.selectEvent.subscribe(handleSelect' , idString , ', 

YAHOO.example.calendar.' , idString , ', true);  

'. 

 

renderCalendarScriptOn: canvas 

 

 calendarScript isNil ifTrue: [^self]. 

 calendarScript nextPutAll: '} 

YAHOO.util.Event.onDOMReady(YAHOO.example.calendar.init); 

'. 

 canvas script: calendarScript contents. 

 calendarScript := nil. 



 

renderDate: idString label: labelString value: aDate callback: aBlock on: 

canvas 

 

 | dateString | 

 dateString := aDate isNil ifTrue: [ 

  nil. 

 ] ifFalse: [(aDate isKindOf: String) ifTrue: [ 

  aDate. 

 ] ifFalse: [ 

  aDate asStringUsingFormat: #(2 1 3 $/ 1 1 ). 

 ]]. 

 canvas fieldSet 

  id: idString , 'Group'; 

  class: 'yui-skin-sam dateGroup'; 

  with: [ 

   canvas label 

    for: idString; 

    class: 'dateLabel'; 

    with: [canvas text: labelString , ':']; 

    yourself. 

   canvas textInput  

    id: idString; 

    class: 'dateInput'; 

    value: dateString;  

    callback: aBlock; 

    yourself. 

   canvas anchor 

    id: 'show' , idString; 

    class: 'calendarAnchor'; 

    with: [ 

     canvas image 

      id: 'calendar'; 

      altText: 'Show Calendar'; 

      url: 'http://192.168.1.2/images/calendar.gif'; 

      yourself. 

    ]; 

    yourself. 

   canvas div 

    class: 'calendarContainer'; 

    id: 'cal' , idString; 

    yourself. 

  ]; 

  yourself. 

 self  

  addCalendarScriptForID: idString 

  title: labelString 

  date: aDate. 



addCalendarScriptForID: idString title: titleString date: aDate 

 

 calendarScript isNil ifTrue: [ 

  (calendarScript := WriteStream on: String new) 

   nextPutAll: ' 

YAHOO.namespace("example.calendar"); 

YAHOO.example.calendar.init = function() { 

'. 

 ]. 

 calendarScript nextPutAll: (self  

  calendarScriptFor: idString 

  title: titleString 

  date: aDate). 



c. Modify the render method to call the new methods: 

renderContentOn: canvas 

 

 canvas form 

  id: 'eventEditor'; 

  with: [ 

   self 

    renderDate: 'eventDate'  

    label: 'Event Date' 

    value: event date  

    callback: [:value | event date: value] 

    on: canvas. 

   canvas label 

    for: 'location'; 

    id: 'locationLabel'; 

    with: 'Location:'; 

    yourself. 

   canvas textInput 

    id: 'location'; 

    value: event location; 

    callback: [:value | event location: value]; 

    yourself. 

   canvas label 

    for: 'description'; 

    id: 'descriptionLabel'; 

    with: 'Description:'; 

    yourself. 

   canvas textArea 

    id: 'description'; 

    value: event description; 

    callback: [:value | event description: value]; 

    yourself. 

   canvas cancelButton 

    id: 'cancel'; 

    callback: [self answer: false]; 

    with: 'Cancel'; 

    yourself. 

   canvas submitButton 

    id: 'submit'; 

    callback: [self answer: true]; 

    with: 'Save'; 

    yourself. 

  ]. 

 self renderCalendarScriptOn: canvas. 

d. Try the application and experiment with the calendar widget. 

 

41. This ends the prepared tutorial. Please feel free to stay and ask questions and experiment! 

 


